

# NAG Fortran Library Routine Document

## F07GDF (SPPTRF/DPPTRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07GDF (SPPTRF/DPPTRF) computes the Cholesky factorization of a real symmetric positive-definite matrix, using packed storage.

### 2 Specification

```
SUBROUTINE F07GDF(UPLO, N, AP, INFO)
ENTRY      spptrf (UPLO, N, AP, INFO)
INTEGER   N, INFO
real    AP(*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine forms the Cholesky factorization of a real symmetric positive-definite matrix  $A$  either as  $A = U^T U$  if UPLO = 'U', or  $A = LL^T$  if UPLO = 'L', where  $U$  is an upper triangular matrix and  $L$  is lower triangular, using packed storage.

### 4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates whether the upper or lower triangular part of  $A$  is stored and how  $A$  is factorized, as follows:

if UPLO = 'U', the upper triangular part of  $A$  is stored and  $A$  is factorized as  $U^T U$ , where  $U$  is upper triangular;

if UPLO = 'L', the lower triangular part of  $A$  is stored and  $A$  is factorized as  $LL^T$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

3: AP(\*) – *real* array *Input/Output*

**Note:** the dimension of the array AP must be at least  $\max(1, N * (N + 1) / 2)$ .

*On entry:* the  $n$  by  $n$  symmetric positive-definite matrix  $A$ , packed by columns. More precisely, if UPLO = 'U', the upper triangle of  $A$  must be stored with element  $a_{ij}$  in AP( $i + j(j - 1)/2$ ) for  $i < j$ ; if UPLO = 'L', the lower triangle of  $A$  must be stored with element  $a_{ij}$  in AP( $i + (2n - j)(j - 1)/2$ ) for  $i > j$ .

*On exit:* the upper or lower triangle of  $A$  is overwritten by the Cholesky factor  $U$  or  $L$  as specified by UPLO, using the same packed storage format as described above.

4: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the leading minor of order  $i$  is not positive-definite and the factorization could not be completed. Hence  $A$  itself is not positive-definite. This may indicate an error in forming the matrix  $A$ . To factorize a symmetric matrix which is not positive-definite, call F07PDF (SSPTRF/DSPTRF) instead.

## 7 Accuracy

If UPLO = 'U', the computed factor  $U$  is the exact factor of a perturbed matrix  $A + E$ , where

$$|E| \leq c(n)\epsilon|U^T||U|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

If UPLO = 'L', a similar statement holds for the computed factor  $L$ . It follows that  $|e_{ij}| \leq c(n)\epsilon\sqrt{a_{ii}a_{jj}}$ .

## 8 Further Comments

The total number of floating-point operations is approximately  $\frac{1}{3}n^3$ .

A call to this routine may be followed by calls to the routines:

F07GEF (SPPTRS/DPPTRS) to solve  $AX = B$ ;

F07GGF (SPPCON/DPPCON) to estimate the condition number of  $A$ ;

F07GJF (SPPTRI/DPPTRI) to compute the inverse of  $A$ .

The complex analogue of this routine is F07GRF (CPPTRF/ZPPTRF).

## 9 Example

To compute the Cholesky factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix},$$

using packed storage.

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07GDF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER       (NMAX=8)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
real           AP(NMAX*(NMAX+1)/2)
*      .. External Subroutines ..
EXTERNAL        spptrf, X04CCF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07GDF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
          READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
          READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
      END IF
*
*      Factorize A
*
      CALL spptrf(UPLO,N,AP,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Print factor
*
          IFAIL = 0
*
          CALL X04CCF(UPLO,'Nonunit',N,AP,'Factor',IFAIL)
*
      ELSE
          WRITE (NOUT,*) 'A is not positive-definite'
      END IF
      END IF
      STOP
*
      END

```

## 9.2 Program Data

```

F07GDF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  4.16
 -3.12   5.03
  0.56  -0.83   0.76
 -0.10   1.18   0.34   1.18   :End of matrix A

```

### 9.3 Program Results

F07GDF Example Program Results

Factor	1	2	3	4
1	2.0396			
2	-1.5297	1.6401		
3	0.2746	-0.2500	0.7887	
4	-0.0490	0.6737	0.6617	0.5347

---